

Dvojný integrál

Robert Mařík


27. listopadu 2010



1 Dvojný a dvojnásobný integrál

Dvojný integrál v systémech počítačové algebry většinou nefiguruje, je nutno jej již zadat jako integrál dvojnásobný. Přitom může dojít k chybě způsobené člověkem. Pro kontrolu, zda nerovnice definují opravdu oblast přes kterou se má integrovat, může posloužit funkce `region_plot`, která vykreslí řešení soustavy nerovnic o dvou neznámých. Metoda `show(aspect_ratio=1)` slouží k tomu, že kružnice nejsou deformované a vypadají jako kružnice (dojde k volbě stejného měřítka na ose x a ose y).

```
Sage code
x,y = var("x,y")
region_plot([0<x, x<1, 1-x<y, y<1-x^2], (x,0,1.2), (y,0,1.2)).show(aspect_ratio=1,figsize=5)
```



Pokud je oblast integrace v pořádku, můžeme vypočítat dvojný intgerál. V následujícím příkladě je množina M zadána jako ohraničená oblast mezi křivkami $y = 1 - x$ a $y = 1 - x^2$ a počítáme integrál

$$\iint_M x^3 dx dy = \int_0^1 \left[\int_{1-x}^{1-x^2} x^3 dy \right] dx.$$

⁰Podporováno grantem FRVŠ 131/2010.

⁰Dílo je šířeno pod licencí Creative Commons: Uveďte autora – neužívejte komerčně.

Sage code

```
integrate(integrate(x^3,y,1-x,1-x^2),x,0,1)
```

$\frac{1}{30}$

Zkusíme ještě druhé pořadí integrace, nejdřív musíme správně zapsat integrační oblast pomocí nerovnic.

Sage code

```
A=solve(1-x==y,x); A
```

$[x = -y + 1]$

Sage code


```
B=solve(1-x^2==y,x); B
```

$[x = -\sqrt{-y+1}, x = \sqrt{-y+1}]$

Vizuální kontrola

Sage code

```
region_plot([0<y, y<1, 1-y<x, x<sqrt(1-y)], (x,0,1.2), (y,0,1.2)).show(aspect_ratio=1, figsize=2)
```



Výpočet

Sage code

```
integrate(integrate(x^3,x,1-y,sqrt(1-y)),y,0,1)
```

Traceback (most recent call last):

```
...  
-- an error. To debug this try: debugmode(true);
```

Chyba?

```
|pre|defint: upper limit of integration must be real; found sqrt(1-y)  
-- an error. To debug this try: debugmode(true);|pre|
```

Ale náš integrál je zapsán správně a pro takovou chybovou hlášku není důvod

I počítačům je někdy nutno pomoci, například přidáme předpoklad, který počítač přesvědčí, že $\sqrt{1-y}$ je reálné číslo. Použijeme příkaz `assume`.

Sage code

```
assume(y<1)  
integrate(integrate(x^3,x,1-y,sqrt(1-y)),y,0,1)
```

$\frac{1}{30}$

Vidíme, že pro opačné pořadí integrace je výsledek stejný. Jiná metoda jak vnutit výraz $\sqrt{1-y}$ do horní meze je použití parametru, za který potom dosadíme $\sqrt{1-y}$. Lze to udělat například následovně.

Sage code

```
b=var('b')  
integrate(integrate(x^3,x,1-y,b).subs(b=sqrt(1-y)),y,0,1)
```

$\frac{1}{30}$

2 Aplikace dvojného integrálu


2.1 Poloha těžiště

Je-li $\sigma(x, y)$ plošná hustota dvourozměrné množiny M , je hmotnost této množiny $m = \iint_M \sigma(x, y) \, dx dy$. Poloha těžiště je $[x_T, y_T]$, kde jednotlivé souřadnice vypočteme z následujících vzorců.

$$x_T = \frac{1}{m} \iint_M x \sigma(x, y) \, dx dy$$
$$y_T = \frac{1}{m} \iint_M y \sigma(x, y) \, dx dy$$

Uřčíme těžiště půlkruhu o poloměru $r = 1$ vyrobeného z homogenního materiálu. Volíme $\sigma(x, y) = 1$ a $M = \{(x, y) : -1 \leq x \leq 1, 0 \leq y \leq \sqrt{1 - x^2}\}$.

```
_____ Sage code _____  
meze=(-1<x,x<1,0<y,y<sqrt(1-x^2))  
P=region_plot(meze,(x,-1,1),(y,0,1),plot_points=500).show(aspect_ratio=1, figsize=3)  
P
```



Vypočteme hmotnost půlkruhu. Výsledek však již známe dopředu! Vzhledem k volbě $\sigma = 1$ bude hmotnost rovna obsahu půlkruhu a ten umíme snadno vypočítat pomocí vzorců ze střední školy. Trik s `assume` známe z předchozího textu.

```
_____ Sage code _____  
assume(1-x^2>0)  
m=integrate(integrate(1,(y,0,sqrt(1-x^2))), (x,-1,1))  
m
```

$$\frac{1}{2} \pi$$

```
_____ Sage code _____  
x_T=integrate(integrate(x,(y,0,sqrt(1-x^2))), (x,-1,1))  
y_T=integrate(integrate(y,(y,0,sqrt(1-x^2))), (x,-1,1))  
teziste = 1/m*x_T, 1/m*y_T  
teziste
```

$$\left(0, \frac{4}{3\pi}\right)$$

Že je x -ová poloha těžiště nulová vidíme ze symetrie. Druhou souřadnici aproximujeme desetinným číslem. Vzhledem k tomu že ve spodní části půlkruhu je více materiálu než nahoře, očekáváme méně než 0.5.


```
_____ Sage code _____  
n(teziste[1])
```

0.424413181578388

Pokusme se ještě zakrestit těžiště do obrázku. Protože grafiku vytvořenou příkazem `region_plot` není možno kombinovat s dalšími grafickými objekty, sestavíme obrázek tak, že nakreslíme horní půlkružnici, vyšrafujeme plochu mezi grafem a osou x a přidáme červený bod v místě těžiště. Použití `show` a `aspect_ratio` zajistí, že kružnice není nijak deformovaná.

Sage code

```
show(plot(sqrt(1-x^2), (x, -1, 1), fill='axis')+point2d(teziste, color='red', size=50), aspect_ratio=1)
```



Pro srovnání, výpočet v polárních souřadnicích. Ten je možno použít, pokud výpočet v kartézských souřadnicích ztroskotává.

Sage code

```
r, phi = var('r phi')
jakobian = r
m = integrate(integrate(1*jakobian, (r, 0, 1)), (phi, 0, pi))
show(m)
x_T = integrate(integrate(r*cos(phi)*jakobian, (r, 0, 1)), (phi, 0, pi))
y_T = integrate(integrate(r*sin(phi)*jakobian, (r, 0, 1)), (phi, 0, pi))
teziste = 1/m*x_T, 1/m*y_T
teziste
```

$$\frac{1}{2} \pi$$

$$\left(0, \frac{4}{3\pi}\right)$$

2.2 Moment setrvačnosti

Moment setrvačnosti hraje roli při výpočtech rotačních pohybů, ale i při výpočtech pevnosti a při dimenzování nosníků a nosných tyčí. Při dimenzování hraje roli tvar průřezu. Momenty setrvačnosti běžných tvarů je možno najít v literatuře, obecný vzorec je

$$J = \iint_M \rho^2(x, y) \sigma(x, y) \, dx dy,$$

kde $\rho(x, y)$ je vzdálenost bodu (x, y) od osy otáčení. Pro kruh o poloměru R , plošné hustotě $\sigma = \frac{m}{\pi R^2}$ (hmotnost vztahovaná na jednotku plochy) a osu procházející středem kolmo na kruh máme $\rho(x, y) = \sqrt{x^2 + y^2}$ a dostáváme následující výsledek.

Sage code

```
R = var('R')
m = var('m') # zapomeneme promennou m
sigma = m/(pi*R^2)
forget() # zapomeneme vsechny predpoklady zadane prikazem assume
assume(R>0, R^2-x^2>0) # sqrt(R^2-x^2) je realne cislo
J = sigma*integrate(integrate(x^2+y^2, (y, -sqrt(R^2-x^2), sqrt(R^2-x^2))), (x, -R, R))
J
```

$$\frac{1}{2} R^2 m$$

To je i vzorec, který je možno najít v literatuře pro kruhové desky a válce. A ještě přidejme analogický výpočet v polárních souřadnicích.

Sage code

```
sigma = m/(pi*R^2)
jakobian = r
forget()
assume(R>0)
J=sigma*integrate(integrate(jakobian * r^2,(r,0,R)),(phi,0,2*pi))
J
```

$$\frac{1}{2} R^2 m$$

To je i vzorec, který je možno najít v literatuře pro kruhové desky a válce. A ještě přidejme analogický výpočet v polárních souřadnicích. Zde platí $\rho = r$.